
// 1. Program and program units

CompilationUnit = { ProgramUnit "." }.

ProgramUnit = ( Module | Definition | Implementation | Object ).
// 2. Modules

Module = MODULE ModuleName [ ImplementationClause ] ";"
           [ ImportDeclaration ]
             ModuleDeclarations
           ( BlockStatement | END ) SimpleName.

ModuleDeclarations = { SimpleDeclaration | NestedUnit ";" }
                     { ProcedureDeclaration | OperatorDeclaration }
                     { ActivityDeclaration }.
NestedUnit = ( Definition | Implementation | Object ).

ImplementationClause = IMPLEMENTS DefinitionName { "," DefinitionName }.
ImportDeclaration = IMPORT Import { "," Import } ";".
Import = ImportedName [ AS ident ].
ImportedName = ( ModuleName | DefinitionName | ImplementationName | NamespaceName ).

// 3. Definitions

Definition = DEFINITION DefinitionName [ RefinementClause ] ";"
               [ ImportDeclaration ]
                 DefinitionDeclarations
             END SimpleName.
RefinementClause = REFINES DefinitionName.

DefinitionDeclarations = { SimpleDeclaration } { ProcedureHeading | ActivitySignature ";" }.
ActivitySignature = ACTIVITY ActivityName [ CommunicationSyntax END SimpleName ] ";".
// Added
CommunicationSyntax = defined in the separate specification.





// Added
// 4. Implementations

Implementation = IMPLEMENTATION ImplementationName ";"
                   [ ImportDeclaration ]
                     Declarations
                   ( BlockStatement | END ) SimpleName.

// 5. Objects

Object = OBJECT [ ObjModifier ] ObjectName [ FormalParameters ] [ ImplementationClause ] ";"
           [ ImportDeclaration ]
             Declarations
           { ActivityDeclaration }
           ( BlockStatement | END ) SimpleName.
ObjModifier = "{" ident "}".     // VALUE or REF; VALUE by default

ActivityDeclaration = ACTIVITY ActivityName  [ ImplementationClause ] ";"


// Corrected
                        Declarations ( BlockStatement | END SimpleName ).
// 6. Declarations

Declarations = { SimpleDeclaration } { ProcedureDeclaration }.
SimpleDeclaration = ( CONST [DeclModifier] { ConstantDeclaration ";" }
                    | TYPE  [DeclModifier] { TypeDeclaration ";" }
                    | VAR   [DeclModifier] { VariableDeclaration ";" }
                    ).
DeclModifier = "{" ident "}". // PUBLIC or PRIVATE or IMMUTABLE

ConstantDeclaration = ident "=" ConstExpression.

ConstExpression = Expression.
TypeDeclaration = ident "=" Type.
VariableDeclaration = IdentList ":" Type.

// 7. Types

Type = ( TypeName [ Width ] | EnumType | ArrayType | ProcedureType | InterfaceType ).
Width = "{" ConstExpression "}".
ArrayType = ARRAY Length { "," Length } OF Type.

Length = ( Expression | "*" ).





// corrected
EnumType = "(" IdentList ")".
ProcedureType = PROCEDURE [ ProcedureTypeFormals ].
ProcedureTypeFormals = "(" [ PTFSection { ";" PTFSection } ] ")" [ ":" FormalType ].

PTFSection = [ VAR ] FormalType { "," FormalType }.
FormalType = { ARRAY OF } ( TypeName | InterfaceType ).
InterfaceType = OBJECT [ PostulatedInterface ].
PostulatedInterface =  "{" DefinitionName { "," DefinitionName } "}".
// 8. Procedures & operators

ProcedureDeclaration = ProcedureHeading [ ImplementationClause ] ";" [ ProcedureBody ";" ].

ProcedureHeading = PROCEDURE [ ProcModifiers ] ProcedureName [ FormalParameters ].

ProcModifiers = "{" ident { "," ident } "}".    // PRIVATE, PUBLIC, SEALED

ProcedureBody = Declarations BlockStatement SimpleName.
FormalParameters = "(" [ FPSection { ";" FPSection } ] ")" [ ":" FormalType ].

FPSection = [ VAR ] ident { "," ident } ":" FormalType.

OperatorDeclaration = OPERATOR [ OpModifiers ] OpSymbol [ FormalParameters ] ";" OperatorBody ";".
OperatorBody = Declarations BlockStatement OpSymbol.
OpModifiers =   "{" ident { "," ident } [ "," Priority ] "}"
             .| "{" Priority "}".







// corrected
Priority = ConstExpression.








// corrected
OpSymbol = string.   // A 1,2,3-character string; the set of possible symbols is restricted

// 9. Statements
StatementSequence = Statement { ";" Statement }.
Statement = [ Assignment
            | ProcedureCall
            | IfStatement
            | CaseStatement
            | WhileStatement
            | RepeatStatement
            | LoopStatement
            | ForStatement
            | AWAIT Expression
            | EXIT
            | RETURN [ Expression ]
            | BlockStatement
            | PutToChannel






// Added
            | GetFromChannel






// Added
            ].
Assignment = Designator ":=" Expression.
ProcedureCall = Designator.
IfStatement = IF    Expression THEN StatementSequence
            { ELSIF Expression THEN StatementSequence }
            [ ELSE  StatementSequence ]
              END.
CaseStatement =  CASE Expression OF
                     Case { "|" Case }
               [ ELSE StatementSequence ]
                 END.
Case = [ CaseLabel { "," CaseLabel } ":" StatementSequence ].
CaseLabel = ConstExpression [ ".." ConstExpression ].
WhileStatement = WHILE Expression DO StatementSequence END.
RepeatStatement = REPEAT StatementSequence UNTIL Expression.
LoopStatement = LOOP StatementSequence END.
ForStatement = FOR ident ":=" Expression TO Expression [ BY ConstExpression ]
               DO StatementSequence END.
BlockStatement = BEGIN [ BlockModifiers ]
                     StatementSequence
                   { ExceptionHandler }
                   [ CommonExceptionHandler ]
                 END.
BlockModifiers = "{" ident { "," ident } "}".    // LOCKED, CONCURRENT, BARRIER

ExceptionHandler = ON ExceptionName { "," ExceptionName } DO StatementSequence.
CommonExceptionHandler = ON EXCEPTION DO StatementSequence.

PutToChannel = [ Channel ] "!" Value.




// added
GetFromChannel = [ Channel ] "?" Receiver.



// added
Value = Expression.








// added
Receiver = Designator.







// added
Channel = Designator.







// added
// 10. Expressions

Expression =   SimpleExpression
                   [ ( "=" | "#" | "<" | "<=" | ">" | ">=" | IN ) SimpleExpression ]
             | Designator IMPLEMENTS DefinitionName.
SimpleExpression = [ "+"|"­" ] Term { ( "+" | "­" | OR ) Term }.
Term = Factor { ( "*" | "/" | DIV | MOD | "&" ) Factor }.
Factor =   number
         | CharConstant
         | string
         | NIL
         | Set
         | Designator
         | NEW TypeName [ "(" ActualParameters ")" ]
         | NEW ActivityInstanceName






// added
         | "(" Expression ")"
         | "~" Factor.
Set = "{" [ SetElement { "," SetElement } ] "}".




// corrected
SetElement = Expression [ ".." Expression ].

Designator = Instance
           | Designator "^"                                        // Dereference
           | Designator "[" Expression { "," Expression } "]"      // Array element
           | Designator "(" [ ActualParameters ] ")"               // Function call
           | Designator "." MemberName                             // Member selector

Instance = ( THIS | InstanceName | DefinitionName "(" InstanceName ")" ).
ActualParameters = Actual { "," Actual }.
Actual = Expression [ "{" [ VAR ] FormalType "}" ].      // Argument with type signature

// 11. Constants

// Replacements: number, integer, real, digit, string, character
// are terminals; therefore they are written with all lower-case letters.

number = integer | real.
integer = digit { digit } | digit { HexDigit } "H".
real = digit { digit } "." { digit } [ ScaleFactor ].
ScaleFactor = "E" ["+" | "­" ] digit { digit }.




// Corrected

HexDigit = digit | "A" | "B" | "C" | "D" | "E" | "F".
digit = "0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9".
CharConstant = """ character """ | "’" character "’" | digit { HexDigit } "X".
string = """ { character } """ | "’" { character } "’".
character = letter | digit | Other.

Other = // Any character from the alphabet except those that are in use…
// 12. Identifiers & names

// Replacements: ident, letter, digit are terminals;
// therefore they are written with all lower-case letters.

ident = ( letter | "_" ) { letter | digit | "_" }.
letter = "A" | ... | "Z" | "a" | ... | "z" | any other "culturally-defined" letter
IdentList = ident { "," ident }.



// Corrected
QualIdent = { ident "." } ident.



// Corrected
DefinitionName     = QualIdent.
ModuleName         = QualIdent.
NamespaceName      = QualIdent.
ImplementationName = QualIdent.
ObjectName         = QualIdent.
TypeName           = QualIdent.
ExceptionName      = QualIdent.
InstanceName       = QualIdent.
ActivityInstanceName = QualIdent.



// Added
ProcedureName = ident.





// Corrected
ActivityName  = ident.





// Added
MemberName    = ( ident | OpSymbol ).


// Corrected
SimpleName    = ident.





// Corrected

